Dérivées directionnelles

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2006. Cette feuille Maple est la seconde d'une série de deux sur les sujets de la dérivée partielle et de la dérivée directionnelle d'une fonction réelle différentiable de deux variables réelles. Chaque document transpose en Maple la présentation habituellement faite en classe de ces deux notions du calcul différentiel. Sur le plan mathématique, la compréhension des différentes étapes dans la définition de la dérivée partielle et directionnelle, exige pour le lecteur une bonne maîtrise de la géométrie vectorielle de la droite et du plan dans l'espace. Et à un niveau autre que purement mathématique, la transposition graphique en Maple de cette notion est l'occasion pour le lecteur d'enrichir ses compétences dans l'utilisation de Maple. Le lecteur tirera donc de ce document plusieurs éléments enrichissants en autant qu'il puisse en faire une lecture attentive et active.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> macro(grispâle = COLOR(RGB, .9523, .9523, .9523)):
  with(plots,contourplot,contourplot3d,display,gradplot,setoptions3d,
  spacecurve,textplot,textplot3d):
  with(plottools,arrow,disk,sphere,translate):
  with(VectorCalculus,Del,DirectionalDiff):
  setoptions3d(size=[300,300],axes=normal,style=patchnogrid,lightmodel=
  none,axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8]);
```

Dérivées directionnelles

Chaque dérivée partielle est, comme nous le verrons un peu plus loin, une dérivée directionnelle particulière. Rappelons que la dérivée partielle de z par rapport à x est obtenue en traitant le taux de variation de la variable z en faisant varier la variable x tout en maintenant constant la variable y (on visualise cette situation sur la trace de la surface d'équation $z = \mathrm{f}(x, y)$ laissée par un plan parallèle à l'axe des x (et à l'axe des z) tandis que la dérivée partielle de z par rapport à y est obtenue en traitant le taux de variation de la variable z en faisant varier la variable y tout en maintenant constant la variable x (on visualise cette situation sur la trace de la surface d'équation $z = \mathrm{f}(x, y)$ laissée par un plan parallèle à l'axe des y (et à l'axe des z). Cela a été longuement présenté dans le premier document « Dérivées partielles ».

La dérivée directionnelle, quant à elle, va traiter du taux de variation de z lorsque nous faisons varier les deux variables à la fois dans une direction donnée. Nous visualiserons cette situation sur la trace de la surface $z = \mathrm{f}(x, y)$ laissée par un plan parallèle à l'axe des z et orientée selon la direction donnée.

Créons la surface paraboloïque elliptique d'équation $z =$

$$f(x, y) = 25 - \frac{x^2}{25} - \frac{y^2}{9} .$$

```
> f:=(x,y)->25-1/25*x^2-1/9*y^2:
  z=f(x,y);
  Surface:=plot3d([x,y,f(x,y)],x=-25..25,y=-3*sqrt(25-1/25*x^2)..3*sqrt
  (25-1/25*x^2),grid=[40,40],
                  orientation=[35,70,0],style=patchcontour,color=
  turquoise):
  display(Surface,view=[-30..30,-20...20,0..30],scaling=unconstrained);
```

$$z = 25 - \frac{x^2}{25} - \frac{y^2}{9}$$



Limitons-nous au premier octant la visualisation de cette surface.

```
> L1:=spacecurve([t,sqrt((625*9-9*t^2)/25),0],t=0..25,linestyle=1,
  color=navy):
  L2:=spacecurve([t,0,f(t,0)],t=0..25,linestyle=1,color=navy):
  L3:=spacecurve([0,t,f(0,t)],t=0..15,linestyle=1,color=navy):
  Surface:=L||(1..3),plot3d([x,y,f(x,y)],x=0..25,y=0..3*sqrt(25-1/25*
  x^2),color=turquoise,transparency=0.2):
  display(Surface,orientation=[20,80,0],scaling=constrained);
```



Considérons la trace sur la surface $z = f(x, y)$ laissée par un plan parallèle **seulement** à l'axe des z.

Soit le plan d'équations paramétriques

$x = 10 + 4r$
$y = 8 - 3r$
$z = s$; $r, s \in \mathbb{R}$.

Pouvez-vous justifier clairement pourquoi ces équations paramétriques sont celles d'un plan seulement parallèle à l'axe des z ?

Superposons dans un même graphique la surface et le plan dans le premier octant.

```
> Plan_Directionnel:=plot3d([10+4*r,8-3*r,s],r=-10/4..10/3,s=0..20,
  color=plum):
  display([Surface,Plan_Directionnel],scaling=constrained,orientation=
  [20,80,0]);
```



Mettons en évidence la trace de ce plan sur la surface $z = \mathrm{f}(x, y)$.

Vous devriez être en mesure de montrer que les équations paramétriques ci-dessous sont celle de la trace.

$x = 10 + 4\,t,$
$y = 8 - 3\,t,$
$z = f(10 + 4\,t, 8 - 3\,t),\ t \in \mathbb{R}.$

```
> Trace_Plan:=spacecurve([10+4*t,8-3*t,f(10+4*t,8-3*t)],t=-10/4..8/3,
  color=navy,thickness=3):
  display([Surface,Plan_Directionnel,Trace_Plan],orientation=[20,80],
  view=[0..25,-5..16,0..25],scaling=constrained);
```



Sur cette trace, considérons le point $\mathrm{P}(10, 8, \mathrm{f}(10, 8))$.

```
> P:=sphere([10,8,f(10,8)],0.25,color=navy):
  PointP:=textplot3d([10+3,8+3,f(10,8)+2,"P(10,8,f(10,8))"],align=
  {above,left},color=navy):
  Seg_vP:=spacecurve([10,8,t*f(10,8)],t=0..1,linestyle=1,color=khaki):
  Plan_dir:=plot3d([10+4*r,8-3*r,s],r=-10/4..8/3,s=0..20,color=
  grispâle):
  Trace_Plan:=spacecurve([10+4*t,8-3*t,f(10+4*t,8-3*t)],t=-10/4..8/3,
```

```
color=navy,thickness=1):
Ligne_1:=spacecurve([0,62/4,t],t=0..20,linestyle=1,color=grey):
Ligne_2:=spacecurve([10+4*t,8-3*t,0],t=-10/4..8/3,linestyle=1,color=
grey):
Base:=P,PointP,Trace_Plan,Plan_dir,Seg_vP,Ligne_1,Ligne_2:
display(Base,orientation=[15,75],axes=framed,view=[0..25,0..16,0..20]
);
```



Illustrons, sur cette trace, le taux de variation directionnel orienté (dans la direction du plan) de la fonction au point P.

Afin de préciser un accroissement des variables x et y dans la direction de cette trace, nous avons à déterminer cette direction. La direction de la trace est l'angle $\alpha \in [0, \pi]$ que fait le plan avec l'axe des x. La direction est alors aussi donnée à l'aide du vecteur unitaire $\mathbf{u} = (\cos(\alpha), \sin(\alpha), 0)$. Dans le plan $z = 0$, les équations paramétriques du plan donne le vecteur $\mathbf{v} = (4, -3, 0)$ comme vecteur directeur de la trace. Alors, soit $\mathbf{u} = \frac{v}{||v||} = \left(\frac{4}{5}, -\frac{3}{5}, 0\right)$ le vecteur unitaire donnant l'orientation avec laquelle sera calculée la dérivée directionnelle.

Obtenons un point Q en donnant un accroissement $h > 0$ *dans le sens du vecteur* **u**. Cet accroissement h amène les accroissements $h\cos(\alpha) = h\,\frac{4}{5}$ et $h\sin(\alpha) = -h\,\frac{3}{5}$ aux variables x et y respectivement.

Pour les besoins de l'illustration, prenons $h = 10$.

```
> h:=10:
Vecteur_Direction:=arrow([10,8,0], [10+h*4/5,8+h*(-3/5),0], .4, 1.75,
.4, color=plum):
Vdir:=textplot3d([15,6,0,"u"],align={above,left},color=navy):
Ligne_Direction:=spacecurve([10+t*h*4/5,8+t*h*(-3/5),0],t=-2..2,
linestyle=1,color=grey):
Q:= sphere([10+h*4/5,8+h*(-3/5),f(10+h*4/5,8+h*(-3/5))],0.25,color=
navy):
PointQ:=textplot3d([10+h*4/5+0.25,8+h*(-3/5)+0.25,f(10+h*4/5,8+h*
(-3/5))+1,"Q"],\
        align={above,left},color=navy):
Seg_vQ:=spacecurve([10+h*4/5,8+h*(-3/5),t*f(10+h*4/5,8+h*(-3/5))],t=
0..1,thickness=1,color=khaki):
display([P,PointP,Q,PointQ,Trace_Plan,Seg_vP,Seg_vQ,\
```

```
        Vecteur_Direction,Vdir,Ligne_Direction],orientation=[15,60],
axes=framed,view=[0..25,0..16,0..20]);
```



Visualisons la sécante passant par les point P et Q.

```
> Seg_H:=spacecurve([10+h*4/5*t,8+h*(-3/5)*t,f(10+h*4/5,8+h*(-3/5))],t=
0..1,thickness=1,color=green):
if f(10+h*4/5,8+h*(-3/5))<=f(10,8) then
    Seg_vP_bis1:=spacecurve([10,8,t*f(10+h*4/5,8+h*(-3/5))],t=0..1,
thickness=1,color=khaki):
else Seg_vP_bis1:=spacecurve([10,8,t*f(10,8)],t=0..1,thickness=1,
color=khaki):
fi:
if f(10+h*4/5,8+h*(-3/5))<=f(10,8) then
    Seg_vP_bis2:=spacecurve([10,8,t],t=f(10+h*4/5,8+h*(-3/5))..f(10,
8),\
             thickness=1,color=green):
else
Seg_vP_bis2:=spacecurve([10,8,t],t=f(10,8)..f(10+h*4/5,8+h*(-3/5)),\
             thickness=1,color=green):
fi:
#Seg_vP_bis2:=spacecurve([10,8,t],t=f(10+h*4/5,8+h*(-3/5))..f(10,8),\
             thickness=1,color=green):
Sécante_PQ:=spacecurve([10+h*4/5*t,8+h*(-3/5)*t,f(10,8)+t*(f(10+h*
4/5,8+h*(-3/5))-f(10,8))],t=-0.5..1.5,thickness=1,color=green):
display([P,PointP,PointQ,Q,Ligne_Direction,Seg_H,Trace_Plan,Seg_vQ,
Sécante_PQ,\
         Seg_vP_bis1,Seg_vP_bis2,Vecteur_Direction,Vdir],\
         orientation=[15,75],axes=framed,view=[0..25,0..16,0..20]);
```



L'accroissement de la fonction f dans le sens du vecteur **u** $= \left(\frac{4}{5}, -\frac{3}{5}, 0\right)$ au point $P(10, 8, \mathrm{f}(10, 8))$ amène les coordonnées suivantes du point Q:

$$x = 10 + \frac{4h}{5}$$

$$y = 8 - \frac{3h}{5}$$

L'accroissement de la fonction f se calculera alors $\mathrm{f}\left(10 + \frac{4h}{5}, 8 - \frac{3h}{5}\right) - \mathrm{f}(10, 8)$.

Dans la direction du vecteur ***u***, faisons tendre le point Q vers le point P en faisant tendre h vers 0 (avec $h > 0$).

```
> h:='h':
  Limit(('f'(10+4/5*h,8-3/5*h)-'f'(10,8))/h,h=0,right)=Limit((f(10+4/5*
  h,8-3/5*h)-f(10,8))/h,h=0,right);
```

$$\lim_{h \to 0+} \frac{f\left(10 + \frac{4h}{5}, 8 - \frac{3h}{5}\right) - f(10, 8)}{h} = \lim_{h \to 0+} \frac{\frac{100}{9} - \frac{\left(10 + \frac{4h}{5}\right)^2}{25} - \frac{\left(8 - \frac{3h}{5}\right)^2}{9}}{h} \qquad \textbf{(2.1)}$$

```
> ``=normal(rhs(%));
```

$$= \lim_{h \to 0+} \left(\frac{32}{75} - \frac{41h}{625}\right) \qquad \textbf{(2.2)}$$

```
> value(%);
```

$$= \frac{32}{75} \qquad \textbf{(2.3)}$$

L'interprétation graphique de la valeur $\frac{32}{75}$ est alors la pente de la tangente à la surface au point P *dans la direction du vecteur* **u** $= \left(\frac{4}{5}, -\frac{3}{5}, 0\right)$. Plus clairement, avec un déplacement (horizontal) dans le *sens du vecteur* **u**, la valeur de la **fonction augmente** à un taux de $\frac{32}{75}$. Donc, en parcourant la "surface" dans le sens du vecteur **u**, **on s'élève sur la surface** f à un rythme de

$\frac{32}{75} \, \frac{\textit{unités d'élévation}}{\textit{unités de parcours directionnel horizontal}}$, c'est-à-dire que l'**on est en pente ascendante** selon l'orientation du vecteur **u**.

Illustrons cette élévation sur la surface. Le vecteur $\left(\frac{4}{5}, -\frac{3}{5}, \frac{32}{75}\right)$ est alors un vecteur directeur de la tangente (pourquoi ?) passant par le point $P(10, 8, \mathrm{f}(10, 8))$. Des équations paramétriques de cette tangente sont alors les suivantes:

$$x = 10 + \frac{4\,t}{5}$$

$$y = 8 - \frac{3\,t}{5}$$

$$z = f(10, 8) + \frac{t \cdot 32}{75} \quad ; t \in \mathbb{R}.$$

(La tangente sera tracée en deux parties afin de mieux illustrer le sens de la dérivée directionnelle)

```
> Tangente1:=spacecurve([10+4/5*t,8-3/5*t,f(10,8)+t*(32/75)],t=0..15,
  linestyle=1,thickness=2,color=orange):
  Tangente2:=spacecurve([10+4/5*t,8-3/5*t,f(10,8)+t*(32/75)],t=-15..0,
  linestyle=2,thickness=1,color=orange):
  display([P,PointP,Ligne_Direction,Trace_Plan,
          Seg_vP,Vecteur_Direction,Tangente1,Tangente2,Vdir],
          orientation=[80,80,0],axes=framed,view=[0..25,0..16,0..25]);
```



Sélectionnez le graphique en cliquant celui-ci avec le bouton gauche tout en maintenant enfoncé le bouton. Donner différente orientation spatiale au graphique afin de mieux percevoir cette tangente dans l'espace. Faites de même avec le graphique suivant.

```
> display([Surface,P,PointP,Ligne_Direction,Trace_Plan,\
          Seg_vP,Vecteur_Direction,Tangente1,Tangente2,Vdir],\
          orientation=[80,80],axes=framed,view=[0..25,0..16,0..25]);
```



Déterminons, $\mathrm{f}'_u(x, y)$, soit l'expression de la dérivée directionnelle de la fonction f dans le sens du vecteur **u**

$$= \left(\frac{4}{5}, -\frac{3}{5}, 0\right),$$

```
> Limit(('f'(x+4/5*h,y-3/5*h)-'f'(x,y))/h,h=0,right)=Limit((f(x+4/5*h,
  y-3/5*h)-f(x,y))/h,h=0,right);
```

$$\lim_{h \to 0+} \frac{f\left(x+\frac{4h}{5}, y-\frac{3h}{5}\right) - f(x,y)}{h} = \lim_{h \to 0+} \frac{-\frac{\left(x+\frac{4h}{5}\right)^2}{25} - \frac{\left(y-\frac{3h}{5}\right)^2}{9} + \frac{x^2}{25} + \frac{y^2}{9}}{h} \quad \textbf{(2.4)}$$

```
> ``=value(rhs(%));
```

$$= -\frac{8x}{125} + \frac{2y}{15} \quad \textbf{(2.5)}$$

Or, ce résultat correspond au produit scalaire du vecteur des dérivées partielles $\left(f_x, f_y, 0\right)$ de la fonction f avec le vecteur **u** $= \left(\frac{4}{5}, -\frac{3}{5}, 0\right)$. Notons qu'on a posé la troisième composante de ce vecteur des dérivées partielles égale à zéro. Ainsi,

```
> u:=<4/5|-3/5|0>;
  v:=<D[1](f)(x,y)|D[2](f)(x,y)|0>;
```

$$u := \begin{bmatrix} \frac{4}{5} & -\frac{3}{5} & 0 \end{bmatrix}$$

$$v := \begin{bmatrix} -\frac{2x}{25} & -\frac{2y}{9} & 0 \end{bmatrix} \quad \textbf{(2.6)}$$

Développons le produit scalaire $v \cdot u$.

```
> `v·u`=v.u;
```

$$v \cdot u = -\frac{8x}{125} + \frac{2y}{15} \quad \textbf{(2.7)}$$

Évaluons ensuite ce produit scalaire avec l'abscisse et l'ordonnée du point P.

```
> eval((2.7),[x=10,y=8]);
```

$$v \cdot u = \frac{32}{75} \quad \textbf{(2.8)}$$

C'est le résultat que l'on a déjà obtenu.

Mais, qu'en est-il de la valeur de la dérivée dans la direction opposée du vecteur **u**, c'est-à-dire dans la direction du vecteur **-u** ? Pas besoin de Maple pour déduire la valeur $-\frac{32}{75}$. Les propriétés de la multiplication scalaire permettent de poser:

```
> `(-u)·v`=-`u·v`;
  ``=-32/75;
```

$$(-u) \cdot v = -u \cdot v$$

$$= -\frac{32}{75} \quad \textbf{(2.9)}$$

Dans ce cas, avec un déplacement (horizontal) dans la direction et le sens du vecteur **-u**, la valeur de la fonction augmente à un taux de $-\frac{32}{75}$. Donc, en parcourant la "surface" selon le vecteur **-u**, on s'élève à un rythme de

$-\frac{32}{75}\ \frac{\textit{unités d'élévation}}{\textit{unités de parcours directionnel à l'horizontal}}$, c'est-à-dire que l'**on est en pente descendante** selon l'orientation du vecteur **-u**.

```
> h:=5:
  Vecteur_Direction:=arrow([10,8,0], [10+h*4/5,8+h*(-3/5),0], .4, 1.75,
  .4, color=plum):
  Vecteur_Direction_op:=arrow([10,8,0], [10+h*(-4/5),8+h*(3/5),0], .4,
  1.75, .4, color=plum):
  Vdir_op:=textplot3d([7,8,0,"-u"],align={above,left},color=navy):
  Tangente1bis:=spacecurve([10+4/5*t,8-3/5*t,f(10,8)+t*(32/75)],t=-15.
  .0,linestyle=1,thickness=2,color=orange):
  Tangente2bis:=spacecurve([10+4/5*t,8-3/5*t,f(10,8)+t*(32/75)],t=0.
  .15,linestyle=2,thickness=1,color=orange):
  display([P,PointP,Ligne_Direction,Trace_Plan,\
          Seg_vP,Vecteur_Direction,Vecteur_Direction_op,Tangente1bis,
  Tangente2bis,Vdir,Vdir_op],\
          orientation=[35,80,0],axes=framed,view=[0..25,0..16,0..25]);
```



```
> display([Surface,P,PointP,Ligne_Direction,Trace_Plan,\
          Seg_vP,Vecteur_Direction,Vecteur_Direction_op,Tangente1bis,
  Tangente2bis,Vdir,Vdir_op],\
          orientation=[35,60,0],axes=framed,view=[0..25.,0..16,0..25]);
```



L'expression de la dérivée directionnelle dans la direction du vecteur **u** peut être rapidement calculée à l'aide de la macro-commande `DirectionalDiff` de la bibliothèque `VectorCalculus`.

```
> `D[u](f)(x,y)`=DirectionalDiff( f(x,y), <4/5,-3/5>, [x,y] );
```

(2.10)

$$D[u](f)(x,y) = -\frac{8x}{125} + \frac{2y}{15} \quad \textbf{(2.10)}$$

Et sa valeur au point P(10,8,f(10,8)) est alors obtenue ainsi:

```
> `D[u](f)(10,8)`=eval(DirectionalDiff( f(x,y), <4/5,-3/5>, [x,y] ),[x=10,y=8]);
```

$$D[u](f)(10,8) = \frac{32}{75} \quad \textbf{(2.11)}$$

En fait, la macro-commande `DirectionalDiff` opère le produit scalaire **v•u** où **v** est le vecteur des dérivées partielles de la fonction f. Les développements précédents ont mis en évidence un vecteur **v** des dérivées partielles. Ce vecteur **v** des dérivées partielles de la fonction f, soit le vecteur $(f_x, f_y, 0)$ est appelé **vecteur gradient** ou, plus simplement, gradient. Ce vecteur est noté ∇f. Le vecteur gradient s'obtient avec la macro-commande `Gradient` de la bibliothèque `VectorCalculus`. La macro-commande `Del` de cette même bibliothèque est aussi la macro-commande donnant le vecteur gradient.

```
> Del(f(x,y),[x,y,z]);
```

$$\left(-\frac{2x}{25}\right)\bar{e}_x + \left(-\frac{2y}{9}\right)\bar{e}_y + (0)\bar{e}_z \quad \textbf{(2.12)}$$

Le vecteur gradient est formulé sous la forme d'une combinaison linéaire des vecteurs de la base du repère cartésien, qui sont e_x, e_y et e_z, soit la base canonique (base orthonormée). Ici, le mécanisme de la simplification automatique a simplifié $0e_z$.

```
> about((2.12));
Vector(3, [-2/25*x,-2/9*y,0], attributes = [vectorfield, coords = cartesian[x,y,z]]
):
  nothing known about this object
```

On voit donc apparaître la composante nulle de la combinaison linéaire précédente. Ainsi, en tout point P(a, b, $f(a, b)$) de la surface, le gradient est un vecteur horizontal. (Gradient associé à une fonction de deux variables)

Pour obtenir le vecteur gradient formulé sur le base des coefficients de cette combinaison linéaire (composantes), nous devons convertir le résultat en un objet Maple de type `Vector`.

```
> Delf:=convert(Del(f(x,y),[x,y,z]),Vector[row]);
```

$$Delf := \begin{bmatrix} -\frac{2x}{25} & -\frac{2y}{9} & 0 \end{bmatrix} \quad \textbf{(2.13)}$$

Représentons le vecteur gradient au point P(10, 8, f(10, 8)).

```
> V_Grad:=arrow([10,8,0], [10,8,0]+convert(eval(Delf,[x=10,y=8]),list),
  .4, 1.75, .4,color=orange):
  V_Grad_cniveau:=translate(V_Grad,0,0,f(10,8)+.01):
  display([Surface,P,PointP,Ligne_Direction,Trace_Plan,\
          Seg_vP,V_Grad_cniveau,Vecteur_Direction,Vdir],\
          orientation=[-150,60],view=[0..25,0..16,0..25],axes=framed);
```



Puisque la troisième composante de ∇f est nulle, il est d'usage en mathématique de traiter le vecteur gradient ∇f d'une fonction f différentiable de deux variables comme un vecteur dans le domaine de la fonction f.

```
> Delf:=convert(Del(f(x,y),[x,y]),Vector[row]);
```

$$Delf := \left[\begin{array}{cc} -\frac{2x}{25} & -\frac{2y}{9} \end{array}\right] \quad \textbf{(2.14)}$$

De même avec le vecteur donnant l'orientation de la dérivée directionnelle, il sera traité comme un vecteur dans le domaine de la fonction. Leur représentation respective dans le plan cartésien est habituellement accompagnée de quelques courbes de niveaux.

```
> P2d:=disk([10,8],0.2,color=navy):
  PointP2d:=textplot([10.5+.5,8+0.5,"P(10,8)"],align={above,right},
  color=navy):
  Courbes_niveaux2d:=contourplot(f(x,y),x=0..20,y=0..15,grid=[15,15],
  contours=[1/3,1,3,6,9,12,15,18,21,24,27],grid=[30,30],coloring = [
  pink,turquoise],filled=true):
  Courbe_niveau2d:=contourplot(f(x,y),x=0..20,y=0..15,grid=[15,15],
  contours=[f(10,8)],color=navy,thickness=2):
  V_Grad2d:=arrow([10,8], [10+(-2*10/25),8+(-2*8/9)], .2, 0.75, .2,
  color=orange):
  Vecteur_Direction_bis2d:=arrow([10,8], [10+(-4/5),8+(3/5)], .2, 0.75,
  .2, color=plum):
  Ligne_Direction2d:=plot([t,-3/4*t+62/4,t=0..21],linestyle=1,
  thickness=0,color=orange):
  display([Ligne_Direction2d,Vecteur_Direction_bis2d,P2d,PointP2d,
  V_Grad2d,Courbes_niveaux2d,Courbe_niveau2d],scaling=constrained,size=
  [300,300]);
```



Dans ce graphique, isolons le point P et la courbe de niveau passant par ce point.

```
> display([Ligne_Direction2d,Vecteur_Direction_bis2d,P2d,PointP2d,
V_Grad2d,Courbe_niveau2d],scaling=constrained,size=[300,300]);
```



Remarquons dans le graphique précédent que la direction dans laquelle la dérivée a été calculée n'est clairement pas une direction orthogonale au vecteur gradient.

En résumé, le calcul de la dérivée directionnelle en tant que produit scalaire $\mathbf{v} \cdot \mathbf{u}$ peut maintenant être reformulé ainsi:

$$D_u f(x, y) = \nabla f(x,y) \cdot \mathbf{u}.$$

Une question qui se pose naturellement est la suivante. Pour une fonction de deux variables, si en un point donné toutes les dérivées directionnelles sont possibles, on peut se demander dans quelle orientation la dérivée atteint-elle sa valeur maximale ? sa valeur minimale ? Ou encore, existe-il une orientation dans laquelle une telle dérivée est nulle ?

Puisque $\| \nabla f \cdot \mathbf{u} \| = \|\nabla f\| \cos(\theta)$ où θ est l'angle entre les vecteurs ∇f et $\mathbf{u}$ ($\mathbf{u}$ étant un vecteur unitaire), on déduit facilement que

- – - la dérivée directionnelle croit plus rapidement dans la direction du vecteur gradient en un point P(a, b). Dans ce cas, la dérivée selon cette orientation vaut

$$f'_u(a, b) = \|\nabla f(a, b)\| \cos(0) = \|\nabla f(a, b)\|$$

- – - la dérivée directionnelle décroit plus rapidement dans l'orientation opposée du vecteur gradient en un point P. Dans ce cas, la dérivée selon cette orientation vaut

$$f'_u(a, b) = \|\nabla f(a, b)\| \cos(\pi) = - \|\nabla f(a, b)\|$$

- – - toute direction $\mathbf{v}$ orthogonale au vecteur gradient est une direction dans laquelle le taux de variation de la fonction est nulle. En effet

$$f'_u(a, b) = \|\nabla f(a, b)\| \cos\left(\frac{\pi}{2}\right) = 0$$

Un dernier élément reste à établir à propos de la direction du vecteur gradient. On admet facilement que lorsqu'il est possible de se déplacer sur la surface à une '*hauteur constante*' $f(x, y) = c$, c'est que l'on se déplace, dans le domaine de la fonction f, tangentiellement à la courbe de niveau $f(x, y) = c$. Alors, selon cette courbe de niveau, en chacun de ses points la dérivée directionnelle dans la direction de la tangente est donc nulle puisque qu'il n'y a ni pente ascendante ni pente descendante. Or, toute direction orthogonale au vecteur gradient est une direction dans laquelle le taux de variation de la fonction est nul. Alors, la direction du vecteur gradient est orthogonale à la tangente au point de tangence, donc orthogonale à la courbe de niveau $f(x, y) = c$.

Dans le plan $z = 0$, représentons, par une droite, la direction perpendiculaire au vecteur gradient.

Obtenons un vecteur directeur du gradient au point d'abscisse $x = 10$ et d'ordonnée $y = 8$.

```
> Dir_gradient:=eval(convert(Del(f(x,y),[x,y,z]),Vector[row]),[x=10,y=
  8]);
```

$$Dir\_gradient := \left[\begin{array}{ccc} -\frac{4}{5} & -\frac{16}{9} & 0 \end{array}\right] \qquad \textbf{(2.15)}$$

Alors, la direction cherchée est $\left(\frac{16}{9}, -\frac{4}{5}, 0\right)$. Alors, une équation vectorielle de la droite perpendiculaire à la direction du vecteur gradient est:

```
> evalm([x,y,z]=[10,8,0]+t*[16/9,-4/5,0]);
```

$$\left[\begin{array}{ccc} x & y & z \end{array}\right] = \left[\begin{array}{ccc} 10 + \frac{16\,t}{9} & 8 - \frac{4\,t}{5} & 0 \end{array}\right] \qquad \textbf{(2.16)}$$

Ainsi, des équations paramétriques de cette droite sont:

$x = 10 + \frac{16\,t}{9}$

$y = 8 - \frac{4\,t}{5}$

$z = 0;\ \ t \in \mathbb{R}$.

```
> Droite_tangente1:=spacecurve([10+16/9*t,8-4/5*t,0],t=-10..16,color=
  orange):
```

Alors,

```
> Pl:=plot3d([10+r,8+s,f(10,8)],r=-15..16,s=-8..8,color=grispâle):
  Courbe_niveau3d:=contourplot3d(f(x,y),x=0..20,y=0..15,grid=[25,25],
                   contours=[f(10,8)],grid=[30,30],color = navy):
  Trajectoire:=spacecurve([50/3*cos(t),10*sin(t),0],t=0..Pi/2,color=
  navy):
  Droite_tangente2:=translate(Droite_tangente1,0,0,f(10,8)):
  display([Pl,Trajectoire,Surface,P,PointP,
           Seg_vP,V_Grad,Courbe_niveau3d,V_Grad,V_Grad_cniveau,
           Droite_tangente1,Droite_tangente2],
          orientation=[-150,60],view=[0..25,0..16,0..25],axes=framed,
  scaling=constrained);
```



Représentons maintenant en 2d la direction perpendiculaire au vecteur gradient.

La direction déjà établie est $\left(\frac{16}{9}, -\frac{4}{5}\right)$. Alors, une équation vectorielle de la droite direction perpendiculaire est:

```
> evalm([x,y]=[10,8]+t*[16/9,-4/5]);
```

$$\left[\begin{array}{cc} x & y \end{array}\right] = \left[\begin{array}{cc} 10 + \frac{16\,t}{9} & 8 - \frac{4\,t}{5} \end{array}\right] \qquad \textbf{(2.17)}$$

Donc, des équations paramétriques de cette droite sont:

$x = 10 + \frac{16\,t}{9}$

$y = 8 - \frac{4\,t}{5}$; $t \in \mathbb{R}$.

```
> Ligne_Direction2d_perpendiculaire:=plot([10+16/9*t,8-4/5*t,t=-10.
.16],thickness=2,color=magenta):
display([Vecteur_Direction_bis2d,Ligne_Direction2d_perpendiculaire,
P2d,PointP2d,V_Grad2d,Courbe_niveau2d],\
          labels=[x,y],scaling=constrained,view=[0..18,0..13],size=
[300,300]);
```



Maple dispose de la macro-commande `gradplot` de la bibliothèque `plots` permettant d'illustrer un champ de gradients. La longueur respective de chaque vecteur gradient représenté correspond à l'évaluation qui en est faite en x et y du domaine de f. Dans un même graphique, superposons le champ de gradients et des courbes de niveaux de la fonction f.

```
> Courbes_niveaux2d:=contourplot(f(x,y),x=-20..20,y=-15..15,contours=
10,coloring=[pink,turquoise],filled=true):
C:=gradplot(f(x,y),x=-20..20,y=-15..15,grid=[10,10],arrows=slim,
color=orange):
display(Courbes_niveaux2d,C,scaling=constrained,size=[300,300]);
```



Comment pouvez-vous expliquer pourquoi la longueur de chaque vecteur gradient représenté diminue au fur et à mesure que l'on se rapproche de l'origine ?